Sucesiones

Los objetivos de esta práctica son:

- Definir sucesiones reales mediante fórmulas explícitas y calcular límites.
- Definir sucesiones mediante reglas de recurrencia; estudiar sus propiedades hasta hallar el límite.
- Analizar diversas órdenes de Maple que afectan a sucesiones.

1. Comentarios

Al definir sucesiones, se debe señalar explícitamente que la variable es un número natural

Ejemplo: la sucesión $sen(2\pi n)$. Este ejemplo es trivial, porque $sen(2\pi n) = 0$ para cualquier número natural n. Y el límite es 0, claro.Veamos (la orden para calcular límites es **limit**):

```
> limit(sin(2*n*Pi),n=infinity);
```

$$-1..1$$

Maple indica que no hay límite (o no lo puede calcular), aunque sí sabe que la sucesión está acotada entre -1 y 1. ¿Qué ocurre? Que Maple no sobrentiende que n sea un número natural.

Y si n es un número real cualquiera, no tiene por qué ser $sen(2\pi n) = 0$ (y el límite tampoco):

```
> sin(2*n*Pi);
```

$$\sin(2\,n\,\pi)$$

Maple no escribe 0 en lugar de $\sin(2n\pi)$, porque la variable n no tiene por qué ser entera.

Si queremos que considere que n es natural, podemos decirlo de esta forma:

```
> assume(n,posint);
```

Hasta que indiquemos otra cosa, Maple considerará que la variable n es un entero positivo:

```
> sin(2*n*Pi);limit(sin(2*n*Pi),n=infinity);
```

$$0$$

$$0$$

Algunas órdenes admiten indicaciones "locales" que solo valen para la orden en que se dan:

```
> simplify(sin(2*k*Pi),assume=posint);simplify(sin(2*k*Pi));
```

$$0$$

$$\sin(2\,k\,\pi)$$

En el primer caso, Maple sabe que k es un número natural; en el segundo, no.

Sucesiones, listas y conjuntos

Podemos definir sucesiones como sigue: definimos $b(n)$, pero indicamos que n debe ser un entero positivo, es decir, un número natural.

```
> b:=(n::posint)->sin(n*Pi/8);
```

$$b := n{::}posint \to \sin\left(\frac{1}{8}\,n\,\pi\right)$$

Maple reserva la palabra "sucesión" (en inglés, "sequence") para lo que nosotros podríamos llamar una sucesión finita. La orden **seq** sirve para crear sucesiones (finitas):

```
> seq(b(n),n=1..10);seq(n^2,n=3..7);
```

$$\sin\left(\frac{1}{8}\,\pi\right), \frac{1}{2}\,\sqrt{2}, \sin\left(\frac{3}{8}\,\pi\right), 1, \sin\left(\frac{3}{8}\,\pi\right), \frac{1}{2}\,\sqrt{2}, \sin\left(\frac{1}{8}\,\pi\right), 0, -\sin\left(\frac{1}{8}\,\pi\right), -\frac{1}{2}\,\sqrt{2}$$

$$9, 16, 25, 36, 49$$

Otros conceptos relacionados con las sucesiones finitas son las listas y los conjuntos. En Maple, una lista se escribe entre corchetes y un conjunto entre llaves. Una lista es una sucesión encerrada entre corchetes (**[,]**). Esto es una lista:

```
> [seq(n^2,n=3..7)];
```

$$[9, 16, 25, 36, 49]$$

Esto, una sucesión finita cuyos elementos no son números sino listas (pares de números):

```
> seq([n^2,n^3],n=4..10);
```

$$[16, 64], [25, 125], [36, 216], [49, 343], [64, 512], [81, 729], [100, 1000]$$

Y una lista cuyos elementos son pares de números:

```
> [seq([n,n^2],n=3..9)];
```

$$[[3, 9], [4, 16], [5, 25], [6, 36], [7, 49], [8, 64], [9, 81]]$$

El conjunto de los valores de $b(n)$, para n=1,2,...,10, no tiene 10 elementos, sino solo 7:

```
> {seq(b(n),n=1..10)};
```

$$\left\{0, 1, -\frac{1}{2}\,\sqrt{2}, \frac{1}{2}\,\sqrt{2}, -\sin\left(\frac{1}{8}\,\pi\right), \sin\left(\frac{1}{8}\,\pi\right), \sin\left(\frac{3}{8}\,\pi\right)\right\}$$

2. Cálculo de límites

1) Calcular el límite de la sucesión $\sqrt{4n^2 - 1} - (2\,n - 1)$

Ya hemos visto que la orden para calcular límites es **limit**:

```
> limit(sqrt(4*n^2-1)-(2*n-1),n=infinity);
```

$$1$$

La orden **Limit**, con L mayúscula, no calcula el límite, sino que solo lo escribe:

```
> Limit(sqrt(4*n^2-1)-(2*n-1),n=infinity);
```

$$\lim_{n \to \infty} \left(\sqrt{4\,n^2 - 1} - 2\,n + 1\right)$$

Sin embargo, podemos pedir a Maple que halle el valor de esa expresión:

```
> value(%);
```

$$1$$

Juntando las órdenes **Limit** y **limit**, presentamos la solución de una forma más legible:

```
> Limit(sqrt(4*n^2-1)-(2*n-1),n=infinity)=
  limit(sqrt(4*n^2-1)-(2*n-1),n=infinity);
```

$$\lim_{n \to \infty} \left(\sqrt{4\,n^2 - 1} - 2\,n + 1\right) = 1$$

2) Calcular el límite de la sucesión $\dfrac{\sqrt{n}}{\sqrt{n + \sqrt{n + \sqrt{n}}}}$.

```
> Limit(sqrt(n)/sqrt(n+sqrt(n+sqrt(n))),n=infinity)=
  limit(sqrt(n)/sqrt(n+sqrt(n+sqrt(n))),n=infinity);
```

$$\lim_{n \to \infty} \frac{\sqrt{n}}{\sqrt{n + \sqrt{n + \sqrt{n}}}} = 1$$

3. Sucesiones recurrentes

3) Sea la sucesión dada por $u(1) = \sqrt{2}$, $u(2) = \sqrt{2+\sqrt{2}}$, $u(3) = \sqrt{2+\sqrt{2+\sqrt{2}}}$, ...
Probar que tiene límite y hallarlo.

La sucesión responde a la regla $u(n) = \sqrt{2+u(n-1)}$, con $u(1) = \sqrt{2}$. De modo que si definimos $f(x) = \sqrt{2+x}$, entonces la regla es $u(n) = f(u(n-1))$.

```
> restart;f:=x->sqrt(2+x);
```

$$f := x \to \sqrt{2+x}$$

Si la sucesión fuera convergente (no sabemos si lo es o no), tomando límite en la regla $u(n) = \sqrt{2+u(n-1)}$ resultaría: $\lim_{n\to\infty} u(n) = \sqrt{2+\lim_{n\to\infty} u(n)}$, $L = \sqrt{2+L}$:

```
> solve({L=f(L)});
```

$$\{L = 2\}$$

Si la sucesión converge, converge a 2. Mediante este procedimiento, definimos la sucesión:

```
> u := proc(n::posint) option remember;
  if n=1 then sqrt(2) else sqrt(2+u(n-1))
  end if
  end proc:
```

Con la orden **option remember**, una vez que Maple haya calculado algunos *u(n)*, recuerda esos valores. La siguiente vez que los necesite no tendrá que calcularlos, y se ahorra tiempo.

Por ejemplo, calculemos u(4) y demos una expresión decimal aproximada, usando **evalf**::

```
> [u(4),evalf(u(4))];
```

$$\left[\sqrt{2+\sqrt{2+\sqrt{2+\sqrt{2}}}},\ 1.990369453\right]$$

Podemos escribir los primeros términos de forma aproximada (con 6 dígitos):

```
> evalf(seq(u(n),n=1..10),6);
```

1.41421, 1.84776, 1.96157, 1.99037, 1.99759, 1.99940, 1.99985, 1.99996, 1.99999, 2.00000

Esto nos sugiere que la sucesión es creciente y su límite es 2. Vamos a comprobarlo.

¿Será creciente? Comparemos $u(n)$ con $u(n-1)$ y con 2. Como $u(n) = f(u(n-1))$, lo que hacemos es comparar $f(x)$ con x y con 2. Dibujemos las gráficas de f, x y la constante 2:

```
> plot([f(x),x,2],x=0..3,color=[red,green,blue]);
```



Si $0 < x < 2$, es $0 < x < f(x) < 2$. Como $u(1) = \sqrt{2}$, deducimos que $0 < u(1) < f(u(1)) < 2$.
Es decir: $0 < u(1) < u(2) < 2$. Aplicamos lo mismo a $u(2)$ y deducimos que $0 < u(2) < u(3) < 2$.
De la misma manera, $0 < u(3) < u(4) < 2$. Resumiendo: $0 < u(1) < u(2) < u(3) < \ldots < 2$.

Como es creciente y está acotada superiormente, tiene límite real. Y en ese caso el límite es 2.

4. Representación gráfica de sucesiones

Vamos a representar gráficamente la sucesión $b(n) = \sin\left(\frac{n\pi}{8}\right)$.

Primero definimos la sucesión, que es una función definida en los números naturales.

```
> restart;
> b:=(n::posint)->sin(n*Pi/8);
```

$$b := n{::}posint \to \sin\left(\frac{1}{8}\, n\, \pi\right)$$

Con la siguiente orden representamos la sucesión finita b(n), n=1,2,...,20. Es decir, dibujamos la lista de pares de puntos (n,b(n)), para n=1,2,...20. Por defecto, Maple une los puntos con rectas.

```
> plot([seq([n,b(n)],n=1..20)]);
```



A continuación representamos la misma sucesión (finita) con diversas variaciones. Para más detalles, se puede consultar la ayuda de Maple (por ejemplo, *Help/Topic Search.../plot*, o bien, poner el cursor en la palabra **plot** e ir a *Help/Help on "plot"*).

```
> plot([seq([n,b(n)],n=1..20)],style=point);
```



```
> plot([seq([n,b(n)],n=1..20)],x=0..25,y=-2..2,style=point,
  symbol=cross,labels=["n","b(n)"]);
```




```
> plot([seq([n,b(n)],n=1..20)],x=0..20,y=-1..1,style=point,
  labels=["n","b"],title="La sucesión b(n)");
```



Representamos gráficamente los primeros términos de $u(1)=\sqrt{2}$, $u(2)=\sqrt{2+\sqrt{2}}$, $u(3)=\sqrt{2+\sqrt{2+\sqrt{2}}}$, ... que hemos visto que converge a 2.

```
> restart;
> u := proc(n::posint) option remember;
  if n=1 then sqrt(2) else sqrt(2+u(n-1))
  end if
  end proc:
> plot([seq([n,u(n)],n=1..50)],style=point);
```



Los primeros términos parecen crecer. Pero veamos con más detalle la gráfica: con la siguiente orden representamos los 20 primeros términos y dibujamos solo una zona próxima a $y=2$.

```
> plot([seq([n,u(n)],n=1..20)],x=1..20,y=1.99..2.01,
  style=point);
```